

Using Recursion to Trace Lineages in the SAS® ODS Styles.Default Template

Perry Watts, SAS User, Elkins Park, PA

ABSTRACT

As a first step towards learning how to generate customized tables in ODS, it is recommended that the Styles.Default template be written out to the SAS LOG. What greets you in 9.1.3 SAS is a 700-line listing of default values for attributes that are building blocks for 150+ styles. That the template listing is overwhelming is an understatement. You still don't have a clue about how styles generally work in ODS, but you realize that a comprehensive guide can only be found in the output you are looking at.

To translate the ODS Styles.Default template into something useful, inheritance has to be fully expressed. Whole lineages of styles can then be traced and presented alongside their associated attributes. With the newly formatted template guide, you can easily learn what is available when ODS-generated tables require a new appearance.

Inheritance falls out when recursion is applied to the Styles.Default template. In this paper, recursion is introduced by showing brief code examples written in macro and PROC SQL. Next, the DATA step that processes Styles.Default recursively is reviewed in detail. Final output is produced in ODS with an HTML destination.

PROBLEM DEFINITION

There is no way to organize the Styles.Default template by lineage. Instead, members of any given lineage are widely scattered throughout the template. For example in Figure 1, lines 164 to 479 from the template must be scanned to locate the five ancestors for the HEADEREMPHASISFIXED style. The only structural rule that can be counted upon is that ancestors precede their descendents in the Styles.Default template.

Figure 1. A partial listing of the ODS Styles.Default template is presented in linear order. Lines 1 to 163 define styles that have no ancestors. Examples include FONTS, GRAPHFONTS and COLOR_LIST. CONTAINER also has no ancestors (the FROM clause is missing) but all styles of interest in this paper inherit directly or indirectly from CONTAINER. Inheritance below is traced by arrows. The highlighted lineage is complete, since HEADEREMPHASISFIXED has no descendents.

```
164 style Container
165     "Abstract. Controls all container oriented elements." /
166     font = Fonts('DocFont')
167     foreground = colors('docfg')
168     background = colors('docbg');
...
414 style Cell from Container
415     "Abstract. Controls general cells.";
...
450 style HeadersAndFooters from Cell
451     "Abstract. Controls table headers and footers." /
452     font = fonts('HeadingFont')
453     foreground = colors('headerfg')
454     background = colors('headerbg');
...
465 style Header from HeadersAndFooters
466     "Controls the headers of a table.";
...
472 style HeaderEmphasis from Header
473     "Controls emphasized table header cells." /
474     foreground = colors('headerfgemph')
475     background = colors('headerbgemph')
476     font = fonts('EmphasisFont');
477 style HeaderEmphasisFixed from HeaderEmphasis
478     "Controls emphasized table header cells. Fixed font." /
479     font = fonts('FixedEmphasisFont');
```

The reason why lineage listing is so difficult to implement is that each style definition appears only once in the Styles.Default template. What is needed is a judicious use of redundancy. In Figure 2, CONTAINER appears four times as it should; once for each lineage that uses it. Recursive processing can be adjusted so that redundancy is inserted into the display where needed.

Figure 2. Four of the 60 Container lineages are derived recursively from the Styles.Default Template.

Lineage	Line#	Style Element
16	164	style Container
	414	style Cell from Container
	450	style HeadersAndFooters from Cell
	465	style Header from HeadersAndFooters
	472	style HeaderEmphasis from Header
	477	style HeaderEmphasisFixed from HeaderEmphasis
26	164	style Container
	214	style Date from Container
	227	style ContentsDate from Date
32	164	style Container
	169	style Index from Container
	250	style IndexProcName from Index
	260	style ContentProcName from IndexProcName
	262	style ContentProcLabel from ContentProcName
60	164	style Container
	305	style TitlesAndFooters from Container
	319	style SystemTitle from TitlesAndFooters

When styles are listed in lineage order, the developer can quickly identify the attributes that are available for modification along with their default values. The 60 Container lineages span lines 164 to 544 in the Styles.Default template. (Graphics lineages do not inherit from Container. Instead they inherit from GraphComponent which has no ancestor).

RECURSION DEFINED

From Tannenbaum[4, p. 96-98] and Wikipedia[6] the definition for recursion contains two major components:

- 1) Recursion solves problems by finding solutions to smaller instances of the same problem. This approach to problem solving is iterative, because it repeats the same process until a predefined condition is met.
- 2) The predefined condition is called the stopping point or "base case" where no reference is made to a smaller instance of the same problem. Otherwise, processing would become infinite.

The trail of recursive processing can be graphed as an upside-down tree where the root or top node represents the base case, and the leaves represent starting points for recursive processing.

Recursion is illustrated below by looking first at two classic examples recreated in SAS. Art Carpenter shows how to write $n!$ (n -factorial) recursively as a macro [1, p. 394], and a highly abbreviated report of the chain of command in President Obama's cabinet is created with a self-join in PROC SQL. The self-join is a limited version of a fully recursive join available in Oracle®.

EXAMPLE #1: THE FACTORIAL FUNCTION

Art Carpenter's algorithm for n -factorial is reproduced in the source code that follows [1, p. 394]:

```
%macro factorial(number);
  %if &number gt 1 %then %eval(&number * %factorial(%eval(&number-1)));
  %else 1;
%mend factorial;


options nosymbolgen mlogic mlogicnest;
%put The factorial of 5 is: %factorial(5);
```

This example shows how recursion works. By creating a function that calls itself, n -factorial is solved by finding solutions to smaller instances of the same problem. 5-factorial in this call = $5 \times 4!(4! = 4 \times 3!)(3! = 3 \times 2!)(2! = 2 \times 1!)(1! = 1)$. Each time FACTORIAL is invoked, NUMBER is successively decremented by 1. When NUMBER reaches 1 the stopping condition kicks in (%else 1) and the function calls execute as $1 \times 2 \times 3 \times 4 \times 5 = 120$.

Carpenter's solution works in SAS, because both FACTORIAL and EVAL are macro functions that have return values. Unlike C, you cannot assign the output of a function call to a variable in SAS. In other words, %let

`FACT=%eval(&number * %factorial(%eval(&number-1)))` doesn't work. Likewise, a variable assignment is not needed in the PUT statement containing the embedded macro call. `%factorial(5)` simply resolves to 120.

The MLOGICNEST option in the code for n-factorial along with a tree containing a single branch in Figure 3 shows how SAS works recursively in macro.



EXAMPLE #2: USING RECURSION TO DISPLAY AN ORGANIZATIONAL HIERARCHY

A self or reflexive join in PROC SQL is used to display the chain of command in an organization. In this example, President Obama's cabinet has been reduced to two secretaries, and each of the two secretaries has just two subordinates; a perfect binary tree! First, the input data set:

```

data govthierarchy;
  length EName Title $15;
  infile cards missover;
  input ENum EName Title MNum ; /* Prefixes 'E'= Employee and 'M' = Manager */
  cards;
  2 Clinton      SecyState      1
  3 SteinBerg    DepSecyState  2
  6 Lynn         DepSecyDefense 5
  1 Obama        President
  4 Mills        ChiefOfStaff  2
  5 Gates        SecyDefense   1
  7 Mullen       JntChiefsStaff 5
run;

```

Unlike *n-factorial*, the input data do not have to be entered in strict order. For example, President Obama's entry appears in the middle of the data set – not in the beginning. However, for the SQL join to work in SAS, employee numbers have to be assigned in “left-branch” order so that output is organized by lineage affiliation. By way of illustration, employee numbers are affixed to the tree displayed in Figure 4. For a discussion of the left-branch tree traversal algorithm, see [5, p. 780-782].


Recursion in this example takes advantage of the fact that managers are also employees. So looking at a smaller instance of the same problem, let's find out who is Steinberg's boss. That would be Secretary of State Clinton. Then who is Clinton's boss? That would be President Obama. Since President Obama represents the “base case”, execution stops for this lineage. When all the “starting points” (Steinberg, Mills, Lynn, and Mullen) are processed, the program terminates.

In PROC SQL below, the input data set, **govtHierarchy**, is being joined with itself. The ORDER clause guarantees that the output in Figure 4 is printed “depth-first” or by lineage.

```
proc sql;
  select LowerH.Enum as EmpNum, LowerH.ENAME as EmpName,
         LowerH.Title as EmpTitle,
         UpperH.Enum as MgrNum, UpperH.ENAME as MgrName,
         UpperH.Title as MgrTitle
  from govtHierarchy as LowerH, govtHierarchy as UpperH
  where LowerH.MNum eq UpperH.ENum
  order by LowerH.ENum;
quit;
```

Figure 4. The chain of command in the Obama administration maps to a binary tree with four lineages and three levels. President Obama as the “base case” occupies the root node. Leaf nodes with Steinberg to Mullen would be identified as “starting points” for recursion.

EmpNum	EmpName	EmpTitle	MgrNum	MgrName	MgrTitle
2	Clinton	SecyState	1	Obama	President
3	SteinBerg	DepSecyState	2	Clinton	SecyState
4	Mills	ChiefOfStaff	2	Clinton	SecyState
5	Gates	SecyDefense	1	Obama	President
6	Lynn	DepSecyDefense	5	Gates	SecyDefense
7	Mullen	JntChiefsStaff	5	Gates	SecyDefense



In contrast to SAS, ORACLE has implemented recursion in a way that does not require an assignment of a pre-ordered identification number or a self-join [3, p. 275-286]. Instead, the FROM keyword associated with a single table is coupled with ORACLE commands START, CONNECT BY, and PRIOR to produce the following:

Manager	Subordinates
Obama	Clinton
Clinton	Steinberg
Clinton	Mills
Obama	Gates
Gates	Lynn
Gates	Mullen

An Oracle-type solution where ordering is determined exclusively by relationships among data elements is required for creating the ODS lineage tracer. In this instance, PROC SQL is replaced with a couple of SET statements having POINT options.

TRACING ODS STYLES BY LINEAGE

Of interest to the user of *n-factorial* is the starting point (5!) or more correctly, its return value of 120. Values for intermediate points are only displayed to show how recursion works behind the scenes. For the *chain of command* problem, however, the entire tree is the object of interest. What the tree provides is a comprehensive view of employee relationships: both by lineage (top down, hierarchical) and by level (across, peer).


For the ODS lineage-tracer, each lineage becomes the object of interest. In "lineage-trace mode", output from Figure 4 would be reconfigured as:

Lineage	Official #1	Official #2	Official #3
1	President Obama	Secretary of State Clinton	Deputy Secretary of State Steinberg
2	President Obama	Secretary of State Clinton	Chief of Staff Mills
3	President Obama	Secretary of Defense Gates	Deputy Secretary of Defense Lynn
4	President Obama	Secretary of Defense Gates	Joint Chiefs of Staff Mullen

The single lineage located in Figure 1 becomes Lineage #16 in the lineage tracer:

Lineage	Node #1	Node #2	Node #3	Node #4	Node #5	Node #6
16	Container	Cell	HeadersAndFooters	Header	HeaderEmphasis	HeaderEmphasisFixed

Recursion in this example takes advantage of the fact that *styleName* and *fromName* parsed from the original style definition in the input data are both styles. Again, with recursion, a lineage is defined by looking at smaller instances of the same problem: Working backwards, as seen by the direction of the arrows in figure 1, style HEADEREMPHASISFIXED inherits attributes **from** HEADEREMPHASIS. Next, HEADEREMPHASIS inherits **from** HEADER, and so on up to CONTAINER which is the base case. When the value for *fromName* is set to CONTAINER, a single lineage is defined, and when the value for *styleName* is set to CONTAINER, the data step terminates. (See Step #3 below for more details). The 60 lineages derived from the CONTAINER style in the ODS Styles.Default template map to the tree displayed in Figure 5.



CREATING THE ODS STYLES LINEAGE-TRACER

To translate the ODS Styles.Default template into a list of stand-alone CONTAINER lineages, five steps must be followed in strict order:

- 1) Read in the raw file, **stylesDefault.txt** and save CONTAINER styles to two SAS data sets: **containerStyles** and **containerAttributes**. **containerStyles** has three variables: *styleNum*, *styleName* and *styleFrom*. *styleName* and *styleFrom* are parsed from the original style definition in **stylesDefault.txt**. For example, *styleName* = **Pages** and *fromName* = **Document** when the style definition = **Pages from Document**. **containerAttributes** also has three variables: *styleName*, *attributeNum* and *attribute*. Separating styles from

attributes simplifies the recursive processing that only has to be applied to **containerStyles**. **containerAttributes** becomes the data source for detail entries displayed in Figure 8 below.

- 2) Sort **containerStyles** in descending order so that the “starting points” or descendents precede their ancestors. The only motivation for the sort is to simplify data processing.
- 3) Create the **paths** data set by applying recursion to **containerStyles**. This data step is fully reviewed in the next section. Variables saved to the **paths** data set include *numLinks* and *link1-link7*. Styles saved in *link1-link7* define a single complete CONTAINER lineage.
- 4) Post-process the **paths** data set to reverse the effects of the descending sort in step #2. *link7* becomes *node1* in the new data set. Then *link6* becomes *node2*, *link5* becomes *node3* and so on. Since most of the lineages contain fewer than seven styles, *node1* typically will be blank. In this case additional processing is required to remove blank entries from the newly created node array. The output data set, **pathsCompress**, contains the re-ordered and compressed values for *node1-node7*.
- 5) Sort **pathsCompress** by *node2 ... node7*. (*node1* is always set to “Container”). Add a lineage number, *linNum*, to the data. The variables in the final data set, **containerLineages**, include *linNum* and *node1-node7*.

STEP#3: CREATE THE PATHS DATA SET WITH RECURSION

To explain the SAS code for recursion, a small input data set with just 8 observations has been derived from the larger **containerStyles** data set. From Step #2 above, the input data are sorted in descending order so that processing can proceed in a simplified top-down fashion. Descendents appear before their ancestors in **rContainer** (check *styleNum*) with the result that the CONTAINER style now appears at the end of the data set. Even with such a small data set, it is difficult to see that three complete lineages will be fully defined with an application of recursion.

Data Set: rContainer

styleNum	styleName	fromName
83	HeaderEmphasisFixed	HeaderEmphasis
82	HeaderEmphasis	Header
80	HeaderFixed	Header
79	Header	HeadersAndFooters
75	HeadersAndFooters	Cell
64	Cell	Container
39	BylineContainer	Container
10	Container	

Key to understanding the data step for step#3 is the double looping in the code that supports two SET statements and their associated stopping conditions. The DO-Loops, SET, and stopping conditions are highlighted in yellow below.

In the outer loop, the input data are processed in sequential order from the first observation to the last where the stopping condition, *styleName* EQ “Container”, is satisfied. Each time a record is read in from the outer-loop SET statement, an inner loop SET statement is conditionally executed to identify a complete lineage. The inner loop starts searching for lineage members at record *i+1* where *i* is reset each time the outer loop executes. Each lineage is fully defined when *fromName* EQ “Container”.

Processing is conditional for lineage definition in the PATHS data step. To prevent redundant partial lineages from being generated, the value for *oldFromName* must not be the same as the value for *link[1]*, derived from the current value for *styleName*. In the code listing below, *oldFromName* entries are highlighted in blue. Interim output in Figure 6 shows how *oldFromName* does its job.

```

/* STEP #3 DATA PROCESSING */
data paths (keep= NumLinks Link1-Link7);
  array link {7} $30 link1-link7;
  length oldFromName $30;
  retain link1-link7 k;
  start=1;
  do i=start to totobs;
    /* OUTER LOOP: READ IN LEAF NODES FIRST */
    set rContainer point=i nobs=totobs;
    /* STOPPING POINT FOR EXECUTION -- WHERE STYLENAME="CONTAINER" */
    if styleName EQ 'Container' then stop;
  else do;
    /* HOUSEKEEPING OCCURS AFTER THE CREATION OF A COMPLETE LINEAGE */
    if i GT start then do;
      oldFromName=link[2];
      do kk=1 to 7;
        link[kk]=' ';
      end;
    end;
  end;
end;

```

```

link[1]=stylename;
link[2]=fromName;
/* FOR LINEAGES WITH ONLY TWO STYLES. */
if fromName eq 'Container' AND stylename NE oldFromName then do;
  NumLinks=2; output;
end;
else do;
  /* EXECUTE THE INNER LOOP FOR LINEAGES HAVING MORE THAN TWO STYLES.
  DATA ARE POLLED IN THE INNER LOOP FROM [CURRENT_REC + 1] TO TOTObs. */
  iplus1=i+1;
  k=2;
  if link[1] NE oldFromName OR i EQ start then do j=iplus1 to totobs;
    /* CREATE A COMPLETE LINEAGE */
    set rContainer point=j nobs=totobs;
    if styleName EQ link[k] then do;
      k=k+1;
      link[k]=fromName;
      if link[k] eq 'Container' then do;
        NumLinks=k; output;
        leave; /* STOPPING POINT FOR LINEAGE DEFINITION -- WHERE FROMNAME="CONTAINER" */
      end; *if CONTAINER;
    end; *if stylename;
  end; *do j (inner loop);
end; *else do recursive search for one entry;
end; *else do go to next entry;
end; *do i (outer loop);
run;

```

Figure 6. Interim output from step #3 and final output from step#5 are displayed. Only the blue complete lineages are written out to the **paths** data set in step #3. Overlooked are partial lineages where *oldFromName* is equal to link1. Also, as expected from the outer loop SET statement, *link1* and *link2* are identical in value to *styleName* and *fromName* in **rContainer**. In step#5 *link1* - *link7* (descendent to ancestor) are replaced with *node1* - *node7* (ancestor to descendent). The final data are sorted alphabetically by *node2*, *node3*, ..., *node7*. *node1* is excluded from the sort, because it always equals "Container".

Output from Step#3							
oldFromName	link1	link2	link3	link4	link5	link6	link7
HeaderEmphasis	#HeaderEmphasisFixed	HeaderEmphasis	Header	HeadersAndFooters	Cell	Container	
Header	#HeaderFixed	Header	HeadersAndFooters	Cell	Container		
Header	=Header	HeadersAndFooters	Cell	Container			
HeadersAndFooters	=HeadersAndFooters	Cell	Container				
Cell	=Cell	Container					
Container	#BylineContainer	Container					

Output from Step#5							
linNum	node1	node2	node3	node4	node5	node6	node7
1	Container	BylineContainer					
16	Container	Cell	HeadersAndFooters	Header	HeaderEmphasis	HeaderEmphasisFixed	
18	Container	Cell	HeadersAndFooters	Header	HeaderFixed		

ODS WITH PROC REPORT IS USED TO CREATE THE LINEAGE TRACER

Screen snapshots of the lineage tracer are displayed in Figures 7 and 8. The complete tracer, created in ODS with output to HTML, is available on the PhilaSUG website. Styles were customized to create the lineage tracer, and the lineage tracer facilitated the customization; perfect circularity! PROC REPORT also played a central role in the tracer's development. The algorithm for inserting internal links comes from *Carpenter's Complete Guide to the SAS® Report Procedure* [2, p. 257-260]. A more detailed discussion about how the tracer is constructed is scheduled for presentation in a follow-up paper, *Keep in Style with the SAS® ODS Styles.Default Lineage Tracer*.

Figure 7. Part of the cover page for the ODS tracer. Code for the three lineages with arrows is presented in Step #3 above. Pressing the green link for lineage #16 brings up the associated list of attributes displayed in Figure 8.

60 Container Lineages in the ODS Styles.Default Template (Abstract Classes are in Blue)

Lineage#	Style#1	Style#2	Style#3	Style#4	Style#5	Style#6	Style#7
1	Container	BylineContainer					
2	Container	Cell	Data	DataEmphasis	DataEmphasisFixed		
3	Container	Cell	Data	DataEmpty			
4	Container	Cell	Data	DataFixed			
5	Container	Cell	Data	DataStrong	DataStrongFixed		
6	Container	Cell	HeadersAndFooters	Caption	AfterCaption		
7	Container	Cell	HeadersAndFooters	Caption	BeforeCaption		
8	Container	Cell	HeadersAndFooters	Footer	FooterEmphasis	FooterEmphasisFixed	
9	Container	Cell	HeadersAndFooters	Footer	FooterEmpty		
10	Container	Cell	HeadersAndFooters	Footer	FooterFixed		
11	Container	Cell	HeadersAndFooters	Footer	FooterStrong	FooterStrongFixed	
12	Container	Cell	HeadersAndFooters	Footer	RowFooter	RowFooterEmphasis	RowFooterEmphasisFixed
13	Container	Cell	HeadersAndFooters	Footer	RowFooter	RowFooterEmpty	
14	Container	Cell	HeadersAndFooters	Footer	RowFooter	RowFooterFixed	
15	Container	Cell	HeadersAndFooters	Footer	RowFooter	RowFooterStrong	RowFooterStrongFixed
16	Container	Cell	HeadersAndFooters	Header	HeaderEmphasis	HeaderEmphasisFixed	
17	Container	Cell	HeadersAndFooters	Header	HeaderEmpty		
18	Container	Cell	HeadersAndFooters	Header	HeaderFixed		

Figure 8. The drill-down from figure 7 brings up a detailed list of associated attributes for lineage #16. The attributes come from the **containerAttributes** data set described earlier in the paper. The full lineage is recapped in the title, and there is a link at the bottom of the page that takes the viewer back to the cover page.

Lineage # 16: Container Cell HeadersAndFooters Header HeaderEmphasis HeaderEmphasisFixed

Style	Default Assignment	Attribute
Container	Abstract. Controls all container oriented elements.	
	font = Fonts('DocFont')	FONT
	foreground = colors('docfg')	FOREGROUND
	background = colors('docbg');	BACKGROUND
Cell	Abstract. Controls general cells.	
HeadersAndFooters	Abstract. Controls table headers and footers.	
	font = fonts('HeadingFont')	FONT
	foreground = colors('headerfg')	FOREGROUND
	background = colors('headerbg');	BACKGROUND
Header	Controls the headers of a table.	
HeaderEmphasis	Controls emphasized table header cells.	
	foreground = colors('headerfgemph')	FOREGROUND
	background = colors('headerbgemph')	BACKGROUND
	font = fonts('EmphasisFont');	FONT
HeaderEmphasisFixed	Controls emphasized table header cells. Fixed font.	
	font = fonts('FixedEmphasisFont');	FONT
Return to Lineage List		

DATA VALIDATION

PROC SQL is applied to the **containerStyles** SAS data set created from **stylesDefault.txt** to check for accuracy. First, a check is made to identify the styles that inherit directly from CONTAINER:

```
proc sql noprint;
  create table chkNode2 as
  select distinct styleName
  from containerStyles
  where fromName eq 'Container'
  order by styleName;
  select count(*) into :nNode2 from chkNode2;
quit;
```

The fourteen styles listed are identical in number and content to those for style #2 on the first page in the HTML output. Matching lineage end-styles (or tree “leaves”) is trickier and requires a subquery:

```
proc sql noprint;
  create table chkLeaves as
  select distinct styleName
  from containerStyles
  where styleName not in
  (select fromName from containerStyles);
  select count(*) into :nLeaves from chkLeaves;
quit;
```

The NOT IN directive to the subquery satisfies the requirement for “leaf” status: end-styles are never ancestors. That means they won’t be listed in *fromName*. The sixty style names produced from the query match the sixty complete lineages found in the lineage tracer.

SUMMARY AND FUTURE DIRECTIONS

Tracking inheritance in the ODS Styles.Default template becomes possible when recursion is used to express 60 lineages derived from the CONTAINER style. Initially, recursion is defined and illustrated by example. Next, the data step that processes Styles.Default recursively is reviewed. After the data step review, several screen snapshots showing portions of the final output produced in ODS with an HTML destination are presented. Finally, PROC SQL is used to check the HTML output for accuracy.

An abstract for a follow-up paper, *Keep in Style with the SAS® ODS Styles.Default Lineage Tracer* has been submitted for presentation at the 2010 NESUG conference. Plans are also being made to update the lineage tracer to accommodate changes in version 9.2.2 SAS.

COPYRIGHT STATEMENT

The paper, *Using Recursion to Trace Lineages in the SAS® ODS Styles.Default Template*, is protected by copyright law. This means if you would like to use part or all of the original ideas or text from the paper in any type of publication you are welcome to do so. All you need to do is to cite the paper. For all uses that result in corporate or individual profit, written permission must be obtained from the author. Conditions for usage have been modified from <http://www.whatiscopyright.org>.

REFERENCES

- [1] Carpenter, Art. *Carpenter's Complete Guide to the SAS® Macro Language: Second Edition*. Cary, NC: SAS Institute Inc., 2004.
- [2] Carpenter, Art. *Carpenter's Complete Guide to the SAS® REPORT Procedure*. Cary, NC: SAS Institute Inc., 2007.
- [3] Koch, George and Kevin Loney. *ORACLE: The Complete Reference, Third Edition*. Berkeley, CA: Osborne McGraw-Hill, 1995.
- [4] Tanenbaum, Aaron M. and Moshe J. Augenstein. *Data Structures Using Pascal*. Englewood Cliffs, NJ: Prentice-Hall, Inc., 1981.
- [5] Watts, Perry and Samuel Litwin. *Using SAS/GRAPH® Software to Create Trees that Model Developmental Biological Phenomena*. Proceedings of the Sas Users Group International Sixteenth Annual Conference. SAS Institute, Inc. Cary, NC, pp. 779-784, 1991.

WEB CITATIONS:

- [6] [http://en.wikipedia.org/wiki/Recursion_\(computer_science\)](http://en.wikipedia.org/wiki/Recursion_(computer_science)). *Recursion (Computer Science): From Wikipedia, the free encyclopedia*.

SAS PAPERS THAT COVER RECURSION [1, p.394]:

Adams, John H. *The Power of Recursive SAS Macros: How Can a Simple Macro do So Much?* Proceedings of the Twenty-Eighth SAS® User Group International Conference, Seattle, WA, 2003, paper #136.

Benjamin, William E. Jr. *A Pseudo-Recursive SAS Macro*. 1999. Online article available at www.sas.com/techsup/download/observations/obswww18/obswww18.pdf.

Rhoades, Stephen. *Recursion? SAS? Let's Fake It*. Proceedings of the Fourteenth Annual NorthEast SAS® Users Group Conference, Baltimore, MD, pp. 652-657, 2001, paper #PS8013.

Ward, David L. *Using Recursion in the SAS System*. Proceedings of the Pharmaceutical SAS Users Group Conference, Boston, MA, pp.97-98, 2001, paper # CC12.

TRADEMARK CITATION

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are registered trademarks or trademarks of their respective companies.

CONTACT INFORMATION

The author welcomes feedback via email at perryWatts@comcast.net.